

Using the Julia package `MotifCluster`

William G. Underwood

June 23, 2023

Contents

1	Introduction	2
2	Building motif adjacency matrices	2
2.1	An example network	2
2.2	Basic motif adjacency matrix construction	2
2.3	Functional and structural motif adjacency matrices	3
2.4	Weighted motif adjacency matrices	3
2.4.1	Mean-weighted instances	3
2.4.2	Product-weighted instances	4
3	Sampling random weighted directed networks	4
3.1	Directed stochastic block models	4
3.1.1	Constant-weighted directed stochastic block models	5
3.1.2	Poisson-weighted directed stochastic block models	5
3.2	Bipartite stochastic block models	5
3.2.1	Weighted bipartite stochastic block models	6
4	Spectral embedding with motif adjacency matrices	6
4.1	Laplacian matrices	6
4.1.1	Combinatorial Laplacian	7
4.1.2	Random-walk Laplacian	7
4.2	Laplace embedding	7
4.3	Motif embedding	7
5	Motif-based spectral clustering	8

1 Introduction

This tutorial demonstrates how to use the Julia package **MotifCluster**. These methods are detailed in the paper *Motif-Based Spectral Clustering of Weighted Directed Networks*, which is available at [arXiv:2004.01293](https://arxiv.org/abs/2004.01293) [1]. The functionality of the **MotifCluster** package falls into a few main categories:

- Building motif adjacency matrices
- Sampling random weighted directed networks
- Spectral embedding with motif adjacency matrices
- Motif-based spectral clustering

This tutorial comprehensively demonstrates all of these functionalities, showcasing the full capability of the **MotifCluster** package. The package can be installed from the Julia General registry with

```
] add MotifCluster
```

We load some other helpful packages for this tutorial and set the random seed:

```
using SparseArrays
using Random
Random.seed!(0)
```

The **MotifCluster** package can then be loaded with

```
using MotifCluster
```

2 Building motif adjacency matrices

The main novelty in the **MotifCluster** package is its ability to build a wide variety of motif adjacency matrices (MAMs), and to do so quickly. There are several options to consider when building an MAM, which are covered in this section.

2.1 An example network

In order to demonstrate the construction of MAMs, we first need a small weighted directed network \mathcal{G}_1 to use as an example. Note that throughout this package we represent networks by their weighted directed adjacency matrices (possibly in sparse form). This means that for use alongside Julia packages such as **Graphs**, one must manually convert between adjacency matrices and **Graphs** objects.

```
G1 = sparse([0 2 0 0; 0 0 2 3; 0 4 0 0; 4 0 5 0])
```

2.2 Basic motif adjacency matrix construction

The `build_motif_adjacency_matrix` function is the main workhorse for building MAMs with **MotifCluster**. Let's use it to build an MAM for the network \mathcal{G}_1 . First we must choose a motif to look for. A full list can be obtained with:

```
motifs = get_motif_names()
println(motifs)

["Ms", "Md", "M1", "M2", "M3", "M4", "M5", "M6", "M7", "M8", "M9", "M10", "M11",
 "M12", "M13", "Mcoll", "Mexpa"]
```

Let's use the 3-cycle motif \mathcal{M}_1 .

```
mam = build_motif_adjacency_matrix(G1, "M1")
display(mam)

4x4 SparseMatrixCSCFloat64, Int64 with 6 stored entries:
.  1.0 .  1.0
1.0 .  .  1.0
.  .  .  .
1.0 1.0 .  .
```

Note that all the entries are zero except for entries $(1,2)$, $(1,4)$, $(2,1)$, $(2,4)$, $(4,1)$, and $(4,2)$. This is because vertices 1, 2 and 4 form an exact copy of the motif \mathcal{M}_1 in the network \mathcal{G}_1 , and the (i,j) th MAM entry simply counts the number of instances containing both vertices i and j .

2.3 Functional and structural motif adjacency matrices

Looking at our example network \mathcal{G}_1 again, you might notice that there is seemingly another instance of the motif \mathcal{M}_1 in our network \mathcal{G}_1 , on the vertices 2, 3 and 4, albeit with an “extra” edge from 2 to 3. The reason for this is that we instructed `build_motif_adjacency_matrix` to look for *structural* motif instances (this is the default). Structural instances require an exact match, with no extra edges. If we want to also include instances which may have extra edges present, we must instead use functional motif instances:

```
mam = build_motif_adjacency_matrix(G1, "M1"; motif_type = "func")
display(mam)

4x4 SparseMatrixCSCFloat64, Int64 with 10 stored entries:
.  1.0 .  1.0
1.0 .  1.0 2.0
.  1.0 .  1.0
1.0 2.0 1.0 .
```

This time we also pick up the 3-cycle on vertices 2, 3 and 4. Vertices 2 and 4 therefore occur in two distinct instances of the motif, and so their motif adjacency matrix entries are equal to two.

2.4 Weighted motif adjacency matrices

Our example network \mathcal{G}_1 has weighted edges, which we have not yet used: so far our MAMs have been simply counting instances of motifs. This is because the default weighting scheme is “unweighted”, assigning every instance a weight of one.

2.4.1 Mean-weighted instances

We could instead use the “mean” weighting scheme, where every instance is assigned a weight equal to its mean edge weight. The (i,j) th MAM entry is then defined as the sum of these instance weights across all instances containing both vertices i and j :

```
mam = build_motif_adjacency_matrix(G1, "M1"; motif_type = "func", mam_weight_type = "mean")
display(mam)

4x4 SparseMatrixCSCFloat64, Int64 with 10 stored entries:
.  3.0 .  3.0
3.0 .  4.0 7.0
.  4.0 .  4.0
3.0 7.0 4.0 .
```

The 3-cycle on vertices 1, 2 and 4 has edge weights of 2, 3 and 4, so its mean edge weight is 3. Similarly the 3-cycle on vertices 2, 3 and 4 has mean edge weight of 4. Vertices 2 and 4 appear in both, so their mutual MAM entries are the sum of these two mean weights, which is 7.

2.4.2 Product-weighted instances

We can also use the “product” weighting scheme, where every instance is assigned a weight equal to the product of its edge weights. The (i, j) th MAM entry is then defined as the sum of these instance weights across all instances containing both vertices i and j :

```
mam = build_motif_adjacency_matrix(G1, "M1"; motif_type = "func", mam_weight_type = "product")
display(mam)
```

```
4x4 SparseMatrixCSCFloat64, Int64 with 10 stored entries:
. 24.0 . 24.0
24.0 . 60.0 84.0
. 60.0 . 60.0
24.0 84.0 60.0 .
```

The 3-cycle on vertices 1, 2 and 4 has edge weights of 2, 3 and 4, so the product of its edge weights is 24. Similarly the 3-cycle on vertices 2, 3 and 4 has product of edge weights of 60. Vertices 2 and 4 appear in both, so their shared MAM entries are the sum of these two product weights, which is 84.

3 Sampling random weighted directed networks

Building adjacency matrices by hand is tedious, so it is useful to have methods for generating the adjacency matrices of networks drawn from some probabilistic model. We use (weighted) directed stochastic block models (DSBMs) and (weighted) bipartite stochastic block models (BSBMs).

3.1 Directed stochastic block models

First let’s sample the adjacency matrix of a DSBM which has two blocks of vertices; the first containing five vertices and the second containing three. We use strong within-block connections, with the diagonal entries of the connection matrix set to 0.9. The between-block connections are weaker, with the off-diagonal connection matrix entries set to 0.2. Note how the resulting adjacency matrix is denser on its diagonal blocks $\{1, \dots, 5\} \times \{1, \dots, 5\}$ and $\{6, \dots, 8\} \times \{6, \dots, 8\}$, and is sparser on its off-diagonal blocks $\{1, \dots, 5\} \times \{6, \dots, 8\}$ and $\{6, \dots, 8\} \times \{1, \dots, 5\}$. The entries which lie exactly on the diagonal will always be zero, since we only consider networks without self-loops.

```
block_sizes = [5, 3]
connection_matrix = [0.9 0.2; 0.2 0.9]
mam = sample_dsbm(block_sizes, connection_matrix)
display(mam)

8x8 SparseMatrixCSCFloat64, Int64 with 32 stored entries:
. 1.0 1.0 . 1.0 . . 1.0
1.0 . 1.0 1.0 1.0 . 1.0 .
1.0 1.0 . 1.0 1.0 . 1.0 .
1.0 1.0 1.0 . 1.0 . . .
1.0 1.0 1.0 1.0 . . 1.0 .
. . 1.0 . . . 1.0 1.0
. 1.0 . 1.0 . 1.0 . 1.0
. . . . . 1.0 1.0 .
```

3.1.1 Constant-weighted directed stochastic block models

The matrix above has binary entries, indicating that it is the adjacency matrix of an unweighted directed network. The `MotifCluster` package also allows sampling of weighted directed networks. The simplest example of this is “constant” weighting, where we simply multiply each block of the adjacency matrix by a constant.

```
weight_matrix = [5 2; 2 5]
mam = sample_dsbm(block_sizes, connection_matrix;
                  weight_matrix = weight_matrix, sample_weight_type = "constant")
display(mam)
```

8x8 SparseMatrixCSCFloat64, Int64 with 35 stored entries:

.	5.0	5.0	5.0	5.0	.	.	2.0
5.0	.	5.0	5.0	5.0	2.0	.	.
5.0	5.0	.	5.0	5.0	.	2.0	.
5.0	5.0	5.0	.	5.0	.	.	2.0
5.0	5.0	5.0	5.0
.	.	2.0	.	2.0	.	5.0	5.0
.	2.0	2.0	.	.	5.0	.	5.0
.	2.0	.	.	.	5.0	5.0	.

3.1.2 Poisson-weighted directed stochastic block models

We can also use weights drawn randomly from a Poisson distribution, where each block in the adjacency matrix has its own mean parameter. This returns an adjacency matrix with weights which could be any natural number, but is equal in expectation to the constant version. Note that in this scheme it is possible for the weight to be zero, removing an edge which might have otherwise been present.

```
mam = sample_dsbm(block_sizes, connection_matrix;
                  weight_matrix = weight_matrix,
                  sample_weight_type = "poisson")
display(mam)
```

8x8 SparseMatrixCSCFloat64, Int64 with 29 stored entries:

.	4.0	3.0	7.0	3.0	.	.	3.0
6.0	.	.	3.0	6.0	.	.	1.0
5.0	4.0	.	.	7.0	.	.	.
5.0	7.0	2.0	.	3.0	.	.	.
7.0	7.0	5.0	11.0	.	.	.	2.0
1.0	4.0	3.0	.
3.0	.	.	.	2.0	5.0	.	.
.	4.0	10.0	.

3.2 Bipartite stochastic block models

The `MotifCluster` package can also be used to sample bipartite networks. The vertices of a bipartite network are partitioned into “source” and “destination” vertices, and edges are only permitted to go from source vertices to destination vertices. Let’s sample a DSBM with a single block of two source vertices and two blocks of destination vertices, with sizes of three and two respectively. We can use a strong connection probability of 0.9 to the first block of destination vertices, and a weaker connection probability of 0.2 to the second.

```

source_block_sizes = [2]
destination_block_sizes = [3, 2]
bipartite_connection_matrix = [0.9 0.2]
mam = sample_bsbm(source_block_sizes, destination_block_sizes,
                  bipartite_connection_matrix)
display(mam)

7x7 SparseMatrixCSCFloat64, Int64 with 7 stored entries:
. . . 1.0 1.0 1.0 .
. . 1.0 1.0 1.0 . 1.0
. . . . . . .
. . . . . . .
. . . . . . .
. . . . . . .
. . . . . . .

```

3.2.1 Weighted bipartite stochastic block models

Similarly to the more general directed stochastic block models, we can also use constant-weighted or Poisson-weighted edges for bipartite stochastic block models.

```

bipartite_weight_matrix = [7 2]
mam = sample_bsbm(source_block_sizes, destination_block_sizes,
                  bipartite_connection_matrix,
                  bipartite_weight_matrix = bipartite_weight_matrix,
                  sample_weight_type = "poisson")
display(mam)

7x7 SparseMatrixCSCFloat64, Int64 with 6 stored entries:
. . 5.0 . 8.0 . 4.0
. . 13.0 12.0 5.0 . .
. . . . . . .
. . . . . . .
. . . . . . .
. . . . . . .
. . . . . . .

```

4 Spectral embedding with motif adjacency matrices

Spectral methods involve performing eigenvalue and eigenvector operations on matrices related to networks. We work here with weighted *undirected* networks (which have symmetric adjacency matrices), because motif adjacency matrices are always symmetric.

4.1 Laplacian matrices

We can construct two types of Laplacian matrix for a network using the `MotifCluster` package. First we create an example of a weighted undirected network \mathcal{G}_2 .

```

G2 = sparse([0 2 0 0; 2 0 4 3; 0 4 0 5; 0 3 5 0])

```

4.1.1 Combinatorial Laplacian

The combinatorial Laplacian of an adjacency matrix G is $L_c = D - G$, where D is the diagonal matrix of weighted vertex degrees:

```
lap = MotifCluster.build_laplacian(G2, "comb")
display(lap)

4x4 SparseMatrixCSCInt64, Int64 with 12 stored entries:
 2 -2 . .
-2 9 -4 -3
. -4 9 -5
. -3 -5 8
```

4.1.2 Random-walk Laplacian

The random-walk Laplacian of an adjacency matrix G is $L_{rw} = I - D^{-1}G$, where D is the diagonal matrix of weighted vertex degrees and I is the identity matrix:

```
lap = MotifCluster.build_laplacian(G2, "rw")
display(lap)

4x4 SparseMatrixCSCFloat64, Int64 with 12 stored entries:
1.0 -1.0 . .
-0.222222 1.0 -0.444444 -0.333333
. -0.444444 1.0 -0.555556
. -0.375 -0.625 1.0
```

4.2 Laplace embedding

Once we have constructed the desired Laplacian matrix, we use it to embed each vertex into \mathbb{R}^l by finding the eigenvectors associated with its first (smallest magnitude) few eigenvalues. Below we use the random-walk Laplacian, and embedding dimension $l = 2$:

```
spectrum = MotifCluster.run_laplace_embedding(G2, 2, "rw")
display(spectrum["vals"])
display(spectrum["vects"])

2-element VectorFloat64:
0.0
0.7883194624496305
4x2 MatrixFloat64:
-0.5 0.931648
-0.5 0.197212
-0.5 -0.197212
-0.5 -0.232912
```

For a random-walk Laplacian, the first eigenvalue is always zero (up to machine precision) and its corresponding eigenvector is constant.

4.3 Motif embedding

Motif embedding is simply the process of building an MAM and performing Laplace embedding with it. As an example we use the `run_motif_embedding` function on the network \mathcal{G}_3 below.

```
G3 = sparse([0 0 0 0; 0 0 2 3; 0 4 0 0; 4 0 5 0])
```

An artifact of building MAMs is that although the original network may be connected, there is no guarantee that the MAM is also connected. Hence the MAM is restricted to its largest connected component before the Laplacian is formed. We observe this with the network \mathcal{G}_3 , in which only three of the four vertices are embedded.

```
spectrum = MotifCluster.run_motif_embedding(G3, "M1", "func",
                                             "unweighted", 2, "rw", true)

display(spectrum["vals"])
display(spectrum["vects"])

2-element Vector{Float64}:
 6.661338147750939e-16
 1.5
3x2 Matrix{Float64}:
-0.57735  0.707107
-0.57735 -0.707107
-0.57735  0.0
```

5 Motif-based spectral clustering

The overall aim of `MotifCluster` is to use motifs for spectral clustering, so now we see how to extract clusters from the motif-based eigenvector embeddings. The `run_motif_clustering` function handles the entire process of building an MAM, restricting it to its largest connected component, performing eigenvector embedding, and extracting clusters. We therefore take the opportunity to showcase the ability of `MotifCluster` to recover the blocks of a DSBM, demonstrating all of the methods outlined in this vignette.

Let's use a DSBM with three blocks of 10 nodes each.

```
block_sizes = repeat([10], 3)
```

We use strong connections of 0.9 within the blocks, and weaker connections of 0.2 between the blocks.

```
connection_matrix = [0.9 0.2 0.2; 0.2 0.9 0.2; 0.2 0.2 0.9]
```

We also set the within-block edges to be Poisson-weighted with mean 30, and the between-block edges to be Poisson-weighted with smaller mean 10.

```
weight_matrix = [30 10 10; 10 30 10; 10 10 30]
G4 = sample_dsbm(block_sizes, connection_matrix;
                 weight_matrix = weight_matrix, sample_weight_type = "poisson")
```

Now we can run the motif-based spectral clustering algorithm on this network with the 3-cycle motif \mathcal{M}_1 . We build a functional MAM, weighting the instances by their mean edge weights, using the sparse formulation. We restrict this MAM to its largest connected component. Then we construct a random-walk Laplacian and embed it using the first four eigenvalues and eigenvectors. Finally we extract three clusters.


```
motif_cluster = run_motif_clustering(G4, "M1";  
                                     motif_type = "func",  
                                     mam_weight_type = "mean",  
                                     num_eigs = 4,  
                                     type_lap = "rw",  
                                     num_clusts = 3,  
                                     restrict = true)
```

We can evaluate the performance by comparing it to the ground-truth labels using the adjusted Rand index:

```
truth = [repeat([1], 10); repeat([2], 10); repeat([3], 10)]  
score = adjusted_rand_index(motif_cluster["clusts"], truth)  
println(score)  
  
1.0
```

A larger value indicates better recovery of the blocks, with a value of one indicating perfect agreement.

References

- [1] W. G. Underwood, A. Elliott, and M. Cucuringu. Motif-based spectral clustering of weighted directed networks. April 2020. [arXiv:2004.01293](#).